

```

;- DL8MCG --- HAM-Radio-Sub-Tone-Filter mit Görtzel-Algorithmus --- auf UniDsp56-Board von DL9GFA -----
dftSubToneTH      equ    0.0003      ; Schwelle = 500 Hz (Defaultwert)
dftSubToneAvg     equ    1000        ; Anzahl der Mittelungen (Defaultwert)

constSubTone_COS  equ    @COS(2*PI*250.3/SAMPLE_FREQ)

                org x:

SubToneAvgCnt     ds      1
SubTone_COS      ds      1
SubTone_Q1       ds      1
SubTone_TH       ds      1
SubTone_AVG      ds      1

                org y:

SubTone_Q2       ds      1

                org p:

;----- Sub-Tone-Filter Init -----
SubTone_Init      move    #>constSubTone_COS,x0      ; Defaultwert Sub-Tone-Frequenz
                  move    x0,x:SubTone_COS
                  move    #dftSubToneTH,x0          ; Defaultwert Sub-Tone-Schwelle
                  move    x0,x:SubTone_TH
                  move    #dftSubToneAvg,x0
                  move    x0,x:SubTone_AVG
SubTone_Init1     bclr   #DSW_ST1,x:DspStatusWord
                  jmp     SubTone_Init2

;----- Subton-Filter 250,3 Hz (Görtzel - Filter) -----
SubTone_Filter    move    x:SubToneAvgCnt,a
                  sub     #1,a
                  jmi    SubTone_Power
                  move    a,x:SubToneAvgCnt

                  move    x:Tx2_AfDacVal,a
                  asr    #8,a,a                      ; AF-Sample durch 2^8 teilen
                  move    y:SubTone_Q2,y0            ; Q2 -> y0
                  sub     y0,a x:SubTone_Q1,x0        ; x(i) - Q2 -> a , Q1 -> x0
                  move    x:SubTone_COS,y0           ; cos(w) -> y0
                  mac     x0,y0,a x0,y:SubTone_Q2     ; Q1 * cos(w) + x(i) - Q2 -> a , Q1 -> Q2
                  mac     x0,y0,a                    ; 2 * Q1 * cos(w) + x(i) - Q2 -> a
                  nop
                  move    a,x:SubTone_Q1              ; 2 * Q1 * cos(w) + x(i) - Q2 -> Q1
                  rts

;----- Bestimmung der Signalleistung -----
SubTone_Power     move    x:SubTone_Q1,x0            ; Q1 -> x0
                  move    y:SubTone_Q2,y0            ; Q2 -> y0
                  mpy    x0,y0,b x:SubTone_COS,x1    ; Q1*Q2 -> b , -cos(w) -> x1
                  mpy    y0,y0,a                    ; Q2^2 -> a
                  mac     x0,x0,a b,y1              ; Q2^2 + Q1^2 -> a , Q1*Q2 -> y1
                  mac     -x1,y1,a                  ; Q2^2 + Q1^2 -cos(w)*Q1*Q2
                  mac     -x1,y1,a                  ; Q2^2 + Q1^2 -2*cos(w)*Q1*Q2
                  move    x:DspStatusWord,x0        ; bisheriger Zustand
                  move    x:SubTone_TH,x1
                  cmp     x1,a
                  move    #>1<<DSW_ST1,a
                  jge     SubTone_Power2
                  move    #>~(1<<DSW_ST1),a
SubTone_Power2    eor     x0,a
                  and     #>1<<DSW_ST1,a
                  jeq     SubTone_Init2
                  bchg    #DSW_ST1,x:DspStatusWord
                  jsr     SendStatusToUc

;----- Initialisierung für den nächsten Mittelungsdurchgang -----
SubTone_Init2     move    x:SubTone_AVG,x0          ; Mittelungszeit
                  move    x0,x:SubToneAvgCnt        ; N = 5555
                  move    #0,x0
                  move    x0,x:SubTone_Q1           ; Q1 = 0
                  move    x0,y:SubTone_Q2           ; Q2 = 0
                  rts

```